

```

1  finish                      ! kilépés minden process-ből
2  /clear                     ! memória törlése
3
4  /cwd, 'D:\Google_Drive\01_education\07_VEA\19-20-1\APDL\feladat2-eromero_cella'
5  *abbr,START,/input,SA-eromero_cella-basic1,ans
6
7  ! ADATBEVITEL !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
8
9  ! Geometriai adatok
10 l=80                        ! az erőmérő cella hossza
11 a=12                        ! az erőmérő cella magassága
12 s=9                         ! furatok középpontjainak távolsága
13 d=10                        ! furatok átmérője
14 b=8                         ! üreg kibővítése egy négyzettel (élhossz: 8mm)
15
16 ! Anyagtulajdonságok
17 E=2e5                       ! rugalmassági modulusz
18 nu=0.3                      ! Poisson tényező
19
20 ! Peremfeltételek
21 l0=20                       ! megfogás hossza
22 l1=30                       ! terhelés hossza
23 n=1                         ! terhelési eset
24
25 ! ELŐKÉSZÍTÉS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
26
27 /prep7                      ! belépés a preprocesszorba
28
29 ! Geometria megrajzolása
30 blc4,0,0,l,a                ! téglalap rajzolása
31 cm,cella,area               ! téglalap elnevezése cellának
32
33 cyl4,l/2-s/2,a/2,d/2        ! baloldali furat megadása
34 cyl4,l/2+s/2,a/2,d/2        ! jobboldali furat megadása
35 blc5,l/2,a/2,b,b            ! üreg kibővítése
36 cmsel,u,cella               ! első téglalap (cella) elrejtése az aktuális elemek közül
37 cm,ureg,area                ! ami most látható (2 furat + négyzet) elnevezése ureg-nek
38
39 allsel                      ! minden kijelölése (újra minden látható lesz)
40 asba,cella,ureg             ! az ureg kivonása a cellából
41
42 ! Geometria előkészítése peremfeltételekhez
43 lsel,s,loc,y,0               ! az alsó vonal kijelölése
44 ldiv,all,l0/l               ! a kijelölt vonal felosztása 20 és 60 mm-es szakaszokra
45
46 lsel,s,loc,y,a               ! a felső vonal kiválasztása
47 ldiv,all,l1/l               ! a kijelölt vonal felosztása 30 és 50 mm-es szakaszokra
48
49 lsel,r,loc,x,l-l1,l          ! a 30 mm-es szakasz kijelölése
50 ldiv,all,,,3                 ! a kijelölt szakasz felosztása 3 egyenlő részre
51
52 ! Anyagtulajdonságok megadása:
53 mp,ex,1,E                   ! rugalmassági modulusz
54 mp,nuxy,1,nu                ! Poisson tényező
55
56 ! Végeelem háló beállítása, elkészítése
57 et,1,182,,,2                ! elemtípus: 182: 2D lineáris, 2:SA
58 mshape,1,2d                 ! elem alakja (0: négyszög, 1: háromszög)
59 esize,5                      ! elemméret
60 amesh, all                   ! összes felület behalozása
61
62 ! Peremfeltételek megadása
63 lsel,s,loc,y,0               ! a befogás helyének kijelölése
64 lsel,r,loc,x,0,l0            ! a befogás helyének kijelölése
65 nsll,s,1                     ! összes csomópont kijelölése az aktív vonalon
66
67 d,all,ux,0                   ! A kijelölt csomópontok megfogása x irányban
68 d,all,uy,0                   ! A kijelölt csomópontok megfogása y irányban
69
70 lsel,s,loc,y,a               ! a megadott terhelési szakasz kijelölése
71 lsel,r,loc,x,l-l1+(n-1)*l1/3,l-l1+n*l1/3

```

```

72 sfl,all,pres,1          ! terhelés elhelyezése (p=1N/mm^2)
73
74 allsel                  ! minden kijelölése (újra minden látható lesz)
75
76 finish
77
78 ! SZÁMÍTÁS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
79 /sol                    ! belépés a solver-be
80 solve                   ! számítás elvégzése
81 finish                  ! kilépés a solver-ből
82
83
84 ! KIÉRTÉKEKLÉS !!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!
85 /post1                  ! belépés a postprocessor-ba
86
87 plnsol,u,sum,2,1.0      ! elmozdulásmező szemléltetése (2: eredeti alak is, 1.0:
    nagyítás)
88
89 /wait,2                 ! várakozás 2 sec-ig
90
91 plnsol, epel, x,2,1.0   ! epsilon_x fajlagos nyúlás szemléltetése
92
93 /wait,2                 ! várakozás 2 sec-ig
94
95 plnsol,s,eqv,2,1.0     ! redukált feszültség szemléltetése
96
97 /status,all
98

```